Calculations PER WHEEL (c) Ether 12/9/2013

```
(%i1) kill(all)$ stardisp:true$
```

Given I, D, G, and eff,
calculate forward from the motor torque to the wheel force:

```
(%i2) Tmozin: Tstall*(I/Istall)$

      Twozin: Tmozin*G*eff$

      Fwoz: Twozin/(D/2)$

      wheel_force = force: Fwoz/16;
```

$$(\%o5)\quad wheel\_force = \frac{eff * Tstall * G * I}{8 * Istall * D}$$

Calculate backward from V to find the relationship
between V and D/G:

```
(%i6) Df: D/12$   Cf: %pi*Df$   Wrpm: 60*(V/Cf)$

      q1: solve(Sfree = Wrpm*G,V)[1];

      V_DG: rhs(q1)$

      DoG_V: rhs(solve(ratsubst(DoG,D/G,q1),DoG)[1])$
```

$$(\%o9)\quad V = \frac{\pi * Sfree * D}{720 * G}$$

Set wheel force equal to traction
and solve for functions of interest:

```
(%i12) force = W*mu$

       ratsubst(Aslip,I,%)$

       DoG_Aslip: rhs(solve(ratsubst(DoG,D/G,%),DoG)[1])$

       q2: DoG_Aslip = DoG_V$

       q2a: solve(q2,V)[1];

       q2b: solve(q2,Aslip)[1];

       q2c: ratsubst(V_DG,V,q2b);
```

$$(\%o16)\quad V=\frac{\pi*Aslip*eff*Sfree*Tstall}{5760*Istall*\mu*W}$$

$$(\%o17)\quad Aslip=\frac{5760*Istall*\mu*V*W}{\pi*eff*Sfree*Tstall}$$

$$(\%o18)\quad Aslip=\frac{8*Istall*\mu*D*W}{eff*Tstall*G}$$

Compute some examples:

```
(%i19) "CIM specs:"$ Istall:133$ Sfree:5310$ Tstall:343.4$

(%i23) mu:1.0$ V:25$ W:40$ eff:1.0$
       ''q2b,numer;
```

$$(\%o27)\quad Aslip=133.7301832717899$$

```
(%i28) mu:1.0$ V:12$ W:37.5$ eff:0.8$
       ''q2b,numer;
```

$$(\%o32)\quad Aslip=75.22322809038181$$

```
(%i33) mu:1.0$ V:10$ W:37.5$ eff:0.8$
       ''q2b,numer;
```

$$(\%o37)\quad Aslip=62.68602340865152$$